
Appendix: Estimating the number of people living with HIV and 

undiagnosed fraction in Spain in 2013 
 

Table A1. Characteristics of National Registry of AIDS and new HIV diagnoses surveillance system in Spain 

National Registry of AIDS HIV new diagnoses surveillance system 

Collects new AIDS cases following European 
case definition 

Collects new HIV diagnosis following European 
case definition 
 

Initiated in 1983 at national level. 
Collects data retrospectively since 1981 

Initiated in 2007 at national level although some 
regional systems started earlier. Collects data 
retrospectively since 2003 in 9 out of 19 
Autonomous Regions in Spain 

Coverage: 100% from the beginning Coverage increased from 34% in 2003 (9 
Autonomous Regions reported data) to 100% in 
2013.  

Compulsory notification  Compulsory notification  

Notifiers are clinicians but data are 
complemented from other data sources 
(mortality registry, clinical records, hospital 
discharge registries) 

Notifiers are clinicians but data are 
complemented from other data sources 
(laboratory registries, clinical records, hospital 
discharge registries) 

Main variables: age, sex, transmission mode, 
country of birth, date of diagnosis of all AIDS 
cases, date of the first positive HIV test 

Main variables: age, sex, transmission mode, 
country of birth, date of diagnosis, CD4 count, 
date of first CD4 count after HIV diagnosis 

Supporting document



2 

 

Table A2. Estimated cumulative deaths, fraction of undiagnosed infections and number of persons living with 

HIV in Spain in 2013 by transmission category. 

Transmission category Figure Estimates  (95% CI) 

MSM Undiagnosed fraction (%) 18.8 (15.0 - 22.8) 

 PLHIV 55079 (52025 - 58487) 

 deaths 7533 

IDU Undiagnosed fraction (%) 3.5 (2.3 - 6.1) 

 PLHIV 23753 (19255 - 28645) 

 deaths 54265 

Heterosexual Undiagnosed fraction (%) 20.1 (15.2 - 26.3) 

 PLHIV 28384 (26091 - 31240) 

  deaths 12989 

Total Undiagnosed fraction (%) 18.0 (14.3 - 22.1) 

 PLHIV 141000 (128000 - 155000) 

  deaths 80390 

Figure A1: Probabilities of diagnosis from CD4 stages over the studied period. 

Figure A2: Estimated proportion of non-late HIV diagnoses coming from each CD4 stage over the 

period 2003-2013. Dots are the proportions observed in the HIV surveillance system. 

Figure A3: Estimated probability of HIV as a cause of death by transmission category. 

Figure A4: Estimation of the annual number of new diagnoses for different Spanish regions. Dots are 

the new diagnoses reported in the HIV surveillance system. 

Figure A5: Proportion of each transmission category among observed deaths in the AIDS surveillance 

system over the period 1985-2013. 

Bayesian formulation of the model:  
 
Step1: Estimation of the annual number of new HIV diagnoses 
 

Let 𝐻𝑟𝑡  denote the number of new HIV diagnoses observed in a transmission category at 

year t and in region r, and �̃�𝑟𝑡   the restriction of  𝐻𝑟𝑡   to people who developed AIDS. These 

numbers are jointly modeled in the following way: 

𝐻𝑟𝑡 = Poisson(𝑁𝑟𝑡𝜌𝑟𝑡), 

�̃�𝑟𝑡 = Poisson(𝑁𝑟𝑡�̃�𝑟𝑡), 

where the offset 𝑁𝑟𝑡  is the general population size, and the rates 𝜌𝑟𝑡 and �̃�𝑟𝑡  vary according to 

the following equations: 

log�̃�𝑟𝑡 = �̃�𝑟 + �̃�𝑡 , 

log𝜌𝑟𝑡 = 𝛽1�̃�𝑟 + 𝑢𝑟  + 𝛽2�̃�𝑡 + 𝑣𝑡 , 

The spatial terms 𝑢𝑟 and �̃�𝑟 are random effects with normal prior 𝑁(0, 𝜏𝑅) and  𝑁(0, �̃�𝑅), 

respectively. The smoothness on temporal terms 𝑣𝑡 and �̃�𝑡 is induced by assuming a second 

order random walk prior: 

𝑣𝑡+2 = 𝑁(2𝑣𝑡+1 − 𝑣𝑡 , 𝜏𝑇), 

�̃�𝑡+2 = 𝑁(2�̃�𝑡+1 − �̃�𝑡 , �̃�𝑇), 

where 𝑣1, 𝑣2, �̃�1 and �̃�2 are treated as fixed unknown parameters with a normal prior 𝑁(0, 𝜏0). 

Step2: Estimation of the HIV infection incidence 
 

As previously, the smoothness over time of the annual HIV incidence ℎ𝑡 and the diagnosis 

probability 𝑑𝑘𝑡 from each CD4 stage (k=1,2,3) was obtained by assuming second order random 

walk priors on this terms: 

ℎ𝑡+2 = 𝑁(2ℎ𝑡+1 − ℎ𝑡 , 𝜏ℎ), 

𝜂𝑡+2 = 𝑁(2𝜂𝑡+1 − 𝜂𝑡 , 𝜏𝜂), 

where 

𝑑𝑘𝑡  = logit(𝑎𝑘 + 𝑏𝑘 𝜂𝑡), 

and 𝑎𝑘, 𝑏𝑘  are unknown parameters with a normal prior 𝑁(0, 𝜏0 ). 

Let 𝑝𝑡𝑘𝑙 denote the one-step transition probability from state 𝑘 to state 𝑙 at time 𝑡 (year) associated 

with the back-calculation model formulated in Figure 1 of the manuscript. Using the notations used 

in this figure, we have that the transition probability matrix 

𝑃𝑡 = (𝑝𝑡𝑘𝑙) =

(

 
 

𝑝𝑡11 𝑞12(1 − 𝑑1𝑡) 0 0 𝑑1𝑡
0 𝑝𝑡22 𝑞23(1 − 𝑑2𝑡) 0 𝑑2𝑡
0 0 𝑝𝑡33 𝑞34(1 − 𝑑3𝑡) 𝑑3𝑡
0 0 0 1 0
0 0 0 0 1 )

 
 

, 

where 𝑝𝑡𝑘𝑘 = 1 − ∑ 𝑝𝑡𝑘𝑙
5
𝑙>𝑘  for 𝑘 = 1,2,3. 

Given this matrix, the expected number (prevalence) 𝐸𝑡𝑘 of persons in state 𝑘 at time 𝑡 is obtained 

using the following recurrence equations [1]: 

𝐸1 = 𝐼1, 

𝐸𝑡 = 𝑃𝑡
′𝐸𝑡−1 + 𝐼𝑡, 𝑡 ≥ 2  

Where 𝑃𝑡
′ is the transpose of 𝑃𝑡 and 𝐼𝑡 is a 5-dimensional column vector with the annual infection 

incidence ℎ𝑡 as the first component, and all other components equal to 0.  

In the same way, the expected number of new arrivals (incidence) 𝑒𝑡𝑘 of persons in state 𝑘 at time 𝑡 

is obtained using the recurrence relations: 

𝑒1 = 𝐼1, 

𝑒𝑡 = �̅�𝑡
′𝐸𝑡−1 + 𝐼𝑡, 𝑡 ≥ 2  

where �̅�𝑡 is the same matrix as 𝑃𝑡 but with zeros in its diagonal. 

From this formulation, the expected proportion of non-late HIV diagnoses coming from each CD4 

stage (𝑘 = 1,2,3) over time is 

𝜃𝑡𝑘 =
𝐸𝑡𝑘𝑑𝑡𝑘
𝑒𝑡5

 

Let denote 𝑟𝑡1, 𝑟𝑡2 and  𝑟𝑡3, the observed numbers of new non-late HIV diagnoses at time 𝑡 with an 

available CD4 count at diagnosis greater than or equal to 500, between 200 and 500, and less than 

200, respectively. Then, it is assumed that  

𝑟𝑡 = Multinomial(𝜃𝑡 , 𝑅𝑡), 

where 𝑅𝑡 = 𝑟𝑡1 + 𝑟𝑡2 + 𝑟𝑡3. 

The expected number of new diagnoses for year 𝑡 is 

𝑛𝑡 =∑𝑁𝑟𝑡𝜌𝑟𝑡
𝑟

. 

Let 𝛼𝑡  denote the corresponding proportion of late HIV diagnoses observed in the surveillance 

system. Expectations 𝑒𝑡
4 and 𝑒𝑡

5 can be approximated by the estimated number of late and 

non-late HIV diagnoses, respectively: 

�̂�𝑡
4 = 𝛼𝑡𝑛𝑡, 

�̂�𝑡
5 = (1 − 𝛼𝑡)𝑛𝑡. 

The two steps described in this Bayesian formulation are connected using an augmented 

model strategy by the creation of faked zero observations: 

0 = (log𝑒𝑡
4 − log�̂�𝑡

4) + 𝜀𝑡
4  

 0 = (log𝑒𝑡
5 − log�̂�𝑡

5) + 𝜀𝑡
5 , 

where 𝜀𝑡
4 and 𝜀𝑡

5 are normal random deviations with arbitrary high precision: 

𝜀𝑡
4 = 𝑁(0, 𝜏𝜀), 

𝜀𝑡
5 = 𝑁(0, 𝜏𝜀). 

Model implementation:  

The following code is an implementation using JAGS [2] of the Bayesian model described in the 

paper to estimate the infection incidence. Flat priors were used for all parameters, except for 

temporal components whose prior distribution was slightly informative in order to obtain a smooth 

estimate of the HIV incidence curve. The estimation was perform using 40 MCMC chains with a burn-

in period of 2,000. Algorithm convergence was monitored using the Gelman-Rubin diagnostic [3]. 
 
data 

Nt,# Number of years  
Nr,# Number of regions  
N, # Dimension of the data: number of years by number of regions (N=Nr*Nt) 
y[N], # Observed/missing annual HIV diagnoses from the HIV surveillance system 
y_s[N], # Observed annual HIV diagnoses from the AIDS surveillance system 
pop[N], # population sizes 
X[Nt,N], # design matrix 
alpha[Nt], # proportion of concurrent AIDS-HIV diagnoses 
zeros[Nt],zeros_s[Nt], # faked zeros observations 
CD4[Nt,3], # annual number of CD4 data by CD4 stage 
nCD4[Nt], # annual number of CD4 data 
prec, ## global precision for the parameters (prec=1e-4) 

variables  

Mu[Nt] # Expected annual number of diagnoses  
r[N], # region index 
t[N], # time index 
u[Nr], # region effects 
v[Nt], # time effects (autoregressive prior) 
mu[N], # expected rates of HIV diagnosis in the HIV surveillance system 
mu_s[N], # expected rates of HIV diagnosis in the AIDS surveillance system 
P[Nt-1,5,5], # Matrix of transition probabilities 
P0[Nt-1,5,5], # Working matrix  
E[Nt,5], # Expected annual prevalence in the different states 
e[Nt,5], # Expected annual incidence in the different states 
d[Nt-1,3], # Annual diagnosis probabilities 
prop[Nt-1,3], # Expected proportion of non-late HIV diagnoses coming from each CD4 
h[Nt], # Expected annual incidence of the HIV infection 
w[Nt], # autoregressive prior for the time variations of the HIV infection  
g[Nt], # autoregressive prior for the time variations of the diagnosis probabilities 

 
model{  
 
    ################################################################################### 
    ################# Step 1: Estimation of the annual number of diagnoses incidence ## 
    ################################################################################### 
     
    for(i in 1:N) { ## Response  
            y[i] ~ dpois(mu[i]) #HIV diagnosis counts in the SINIHIV  
            y_s[i] ~ dpois(mu_s[i]) #HIV diagnosis counts in the RNS 
        } 
    for(i in 1:N) { ## Expected response 
           log(mu[i]) <- log(pop[i]) + u[r[i]] + v[t[i]] + beta.r*u_s[r[i]] + beta.t*v_s[t[i]] 
           log(mu_s[i]) <- log(pop[i]) + u_s[r[i]] + v_s[t[i]] #  
        } 
 
    for(k in 1:Nr){ ##  prior for region effects  
        u[k] ~ dnorm(0,tau.r)  
        u_s[k] ~ dnorm(0,tau_s.r) 
    } 
    for(l in 3:Nt) { ## prior for period effects (2nd order random walk) 
        v[l]  ~ dnorm(v[l-1]*2-v[l-2], tau.t)  
        v_s[l]  ~ dnorm(v_s[l-1]*2-v_s[l-2],tau_s.t)  
    } 
    ## Priors for the parameters 
    v[1] ~ dnorm(0,prec)  
    v[2] ~ dnorm(0,prec) 
    v_s[1] ~ dnorm(0,prec)  
    v_s[2] ~ dnorm(0,prec)   
    beta.r ~ dnorm(0,prec) 
    tau.r ~ dgamma(1,prec) 
    tau_s.r ~ dgamma(1,prec)  
    beta.t ~ dnorm(0,prec) 
    tau.t ~ dgamma(1,prec) 
    tau_s.t ~ dgamma(1,prec) 
     
     
     
    ################################################################################### 
    ################# Step 2: Estimation of infection incidence ####################### 
    ################################################################################### 
     
    Mu <- X.t %*% mu #annual number of HIV diagnoses across the country  
     
    # Faked zero observations to link the two steps 
    for (i in 1:Nt){ 
     zeros[i] ~ dnorm(log(e[i,5]+eps)-log(Mu[i]*(1-alpha[i])+eps),1e3)#non late HIV diagnosis 
     zeros_s[i] ~ dnorm(log(e[i,4]+eps)-log(Mu[i]*alpha[i]+eps),1e3)#concurrent HIV-AIDS diag. 
     }  
    for(i in 1:(Nt-1)) { 
        CD4[i+1,] ~ dmulti(prop[i,],nCD4[i+1]) 
        } 

    eps <- 1e-3  #to avoid numerical error with the log  
     
    # Expected proportion of non late HIV 
    prop <- E[1:(Nt-1),1:3] * d + eps 
     
    # Expected prevalence and incidence 
    E[1,] <- rep(0,5) 
    e[1,] <- rep(0,5) 
    h[1] <- exp(w[1]) 
     
    for(i in 2:N){ 
        E[i,] <- t(P[i-1,,])%*%E[i-1,] + f*h[i] 
        e[i,] <- t(P0[i-1,,])%*%E[i-1,] + f*h[i] 
        h[i] <- exp(w[i]) 
    } 
     
    # Fraction of recent infection to CD4 stages   
    #f <- c(.75,.25,0,0,0) ## Cori (2015) 
    f <- c(1,0,0,0,0) ## Aalen (1997) 
     
     
    # output: PLHIV and undiagnosed fraction (F)  
    PLHIVD <- (sum(e[,4])+sum(e[,5])-deaths) 
    PLHIV <- (sum(e[,1])-deaths) 
    F<-(1-PLHIVD/PLHIV)*100 
     
    # Transition Probability Matrix  
    for(i in 1:(Nt-1)){ 
        for(col in 1:4){ for(row in (col+1):5){ P[i,row,col] <- 0 }} 
        for(col in 1:5){ for(row in col:5){ P0[i,row,col] <- 0 }} 
        P[i,1,2] <- q[1]*(1-d[i,1]) 
        P[i,1,3] <- 0 
        P[i,1,4] <- 0 
        P[i,1,5] <- d[i,1] 
        P[i,2,3] <- q[2]*(1-d[i,2]) 
        P[i,2,4] <- 0 
        P[i,2,5] <- d[i,2] 
        P[i,3,4] <- q[3]*(1-d[i,3]) 
        P[i,3,5] <- d[i,3] 
        P[i,4,5] <- 0 
        for(k in 1:5){ P[i,k,k] <- 1 - sum(P[i,k,index[k,]]) } 
        for(row in 1:4){ for(col in (row+1):5){ P0[i,row,col] <- P[i,row,col] }} 
    } 
     
     
    #Diagnosis probabilities equation 
    for(i in 1:(n0-1)){ d[i,]= rep(0,3) } #before 1984       
    for(i in n0:(Nt-1)) { 
        for(k in 1:3) {      
                logit(d[i,k]) <- a[k]+b[k]*(g[i-n0+1]-mg)            
                }}       
     
     
    #Probabilities of natural progression  
    q=c(1/5.5,1/4,1) ## Aalen (1997) 
    #q=c(1/3.32,1/8.20,1) ## Cori (2015)  
     
        
    # Prior on HIV incidence curve (2nd order random walk) 
    for (i in 3:Nt){         
        w[i] ~ dnorm(diff.w[i],tau.w)  
        diff.w[i] <- w[i-1]*2-w[i-2]  
    } 
      
    # Prior on diagnosis probabilities variations  (2nd order random walk)  
    mg=mean(g[1:(Nt-n0)])  
    for (i in 3:(Nt-n0)){  
         g[i] ~ dnorm(diff.g[i],tau.g)  
         diff.g[i] <- g[i-1]*2-g[i-2] #RW2 

      }  
     
    # initial values 
    w[1] ~ dnorm(0,prec) 
    w[2] ~ dnorm(0,prec) 
    g[1] ~ dnorm(0,prec)  
    g[2] ~ dnorm(0,prec)   
     
    for(k in 1:3){ 
        b[k] ~ dnorm(0,prec) 
        a[k] ~ dnorm(0,prec) 
        } 
         
    # smooth precision 
    tau.w ~ dgamma(10,prec) 
    tau.g ~ dgamma(10,prec)  
} 

Fig. 1: Multi-state model for the disease progression from HIV infection to diagnosis 

from Sweeting et al. [6]. 

Fig. 2: Overall annual new cases of HIV infections and HIV diagnoses (Spain, 1977-

2013). 

Fig. 3: Annual new cases of HIV infections by transmission category (Spain, 1977-

2013). 

 

Fig3



 

Fig. 4: Annual new HIV diagnoses estimates (in log scales) over the 1984-2013 

period, in the HIV surveillance system (SINIVIH) and in the AIDS surveillance system 

(RNS). 

Fig. 5: Annual new HIV diagnoses estimates by transmission category (Spain, 1984-

2013). MSM indicates men who have sex with men; IDU, injection drug users; 

Other/NA, haemophiliac/transfusion recipient, mother-to-child or unknown 

categories. 

